


Agilent E1960A GSM Mobile Test Application


Product Note



Control Level

Average Power	Control Level
33.88	33
32.469	32
30.466	30
28.499	28
26.486	26
24.601	24
22.667	22
20.77	20
18.776	18

Using the Dynamic Power Measurement to Measure Mobile Station Power Control Levels When Using SACCH TX Level Signaling



This product note presents a programmatic technique for using the Dynamic Power measurement to measure a GSM mobile station's power control level capability when using SACCH (Slow Associated Control Channel) TX Level Signaling. The Dynamic Power measurement and SACCH TX Level Signaling feature are available in the Agilent Technologies E1960A GSM mobile test application, revision A.03.04 (and higher). This technique is based upon the RF power control characteristics specified in 3GPP TS 05.08 section 4.7 version 8.9.0 Release 1999.

This product note is based on another product note: *"Measuring Mobile Station Power Control Levels Using SACCH TX Level Signaling"* (5980-1470E), which describes how to use a Transmit Power measurement loop to measure each of the power control levels. Using the Dynamic Power measurement for this application, rather than the Transmit Power measurement method enables a faster test time and a technique that is not as dependent on the speed of the computer controlling the test set.

Programming examples presented in this product note are written in the HP BASIC programming language, also known as RMB or Rocky Mountain BASIC. The use of HP BASIC is not an endorsement of the HP BASIC product.

Description of SACCH TX Level Signaling

Refer to Appendix B for a review of SACCH messaging.

Prior to the introduction of the SACCH TX Level Signaling feature into the E1960A GSM mobile test application at revision A.03.04, the only method of commanding the mobile station to use a different transmit power level was a channel assignment via the FACCH (Fast Associated Control Channel). The SACCH TX Level Signaling feature now provides a second method that uses SACCH messaging. When SACCH messaging is used and a command to change the transmit power level is received, the mobile station changes its transmit power level in accordance with 3GPP TS 05.08 section 4.7 version 8.9.0 Release 1999 (RF power control). The table below compares the two methods.

Description of the Dynamic Power measurement

The Dynamic Power measurement was introduced into the E1960A GSM mobile test application at revision A.03.04. The Dynamic Power measurement performs a series of consecutive power measurements on a mobile station returning a power measurement and an integrity value for each burst measured. The measurement can be made on any number between 1 and 999 bursts and results are fetched in ranges, each of which returns 100 bursts. Dynamic Power is not a 3GPP specified measurement, and is only available via the test set's remote user interface.

The only trigger source that the Dynamic Power measurement supports is RF Rise.

Method	Speed of change from old_level to new_level	Transition of power level
Channel assignment via FACCH	Fast (≈ 0.5 sec)	Single step old_level to new_level
Signaling using SACCH messaging	Slow (≈ 2.0 sec)	Ramp old_level to new_level in 2 dB steps

Channel assignment via the FACCH is normally used in the manufacturing environment to change a mobile station's transmit power level because of its speed. However, the mobile station's conformance to 3GPP TS 05.08 section 4.7 version 8.9.0 Release 1999 (RF power control) cannot be measured using the FACCH method. The SACCH TX Level signaling feature was introduced to enable this characteristic of a mobile station's performance to be measured.

Measuring the mobile station's power control level steps

Using a sufficiently fast RF power meter it is possible to measure a mobile station's complete power level control range by changing the mobile station's transmit power level using SACCH messaging. The Agilent Technologies 8960's power meter is fast enough to measure, with specified accuracy, each 2 dB step as the mobile changes its transmit power level. Refer to Appendix C for a review of mobile station power control.

Measurement algorithm

A procedure that can be used to measure a mobile station's power control level steps is described below. Prior to executing the procedure, establish a call with the mobile station and turn off all measurements except the Dynamic Power measurement. The procedure measures power control levels from high to low. A low to high procedure could also be implemented. This procedure is represented in flowchart form in figure 1.

Refer to Appendix A for details of test set considerations such as things you should know, unexpected behaviors, and potential problems when using SACCH MS TX Level Signaling.

1. Set the Expected Power to MS TX Level start_level.
2. Set the Trigger State for the Dynamic Power measurement to Single.
3. Set the Expected Maximum Difference from Previous Measurement parameter for the Dynamic Power Measurement.
4. Set the number of bursts for the Dynamic Power measurement to 300 (high enough to capture all 10 power level steps).
5. Set MS TX Level FACCH Signaling = ON.
6. Set the mobile station's MS TX Level to start_level.
7. Set MS TX Level FACCH Signaling = OFF (this leaves only SACCH TX level signaling on).
8. Set the mobile station's MS TX Level to stop_level.
9. Initiate the Dynamic Power measurement. Fetch the three ranges (3 x 100 bursts) of Dynamic Power measurement and integrity results.
10. Store the results in arrays for post processing.
11. Check the integrity value for each burst measured and exit if not zero.




Figure 1. Flowchart for Measuring Power Control Levels

Enabling and Disabling FACCH Signaling Programmatically

Enabling and disabling TX Level FACCH Signaling is done using a CALL Processing subsystem command. The command is:

```
CALL:SIGNaling:MS:TXLevel:FACCH<sp>1|ON|0|OFF
```

When TX Level FACCH Signaling is set to ON (1), the base station emulator uses both a FACCH channel assignment and an update to the SACCH header to signal the mobile to change to a new power level.

When TX Level FACCH Signaling is set to OFF (0), the base station emulator uses only an update to the SACCH header to signal the mobile to change to a new power level. A FACCH channel assignment message is not sent.

Example Program for Measuring a Mobile Station's Power Control Levels

The following example program assumes that an active link is established between the test set and the mobile station (that is, the test set is in active cell operating mode and the call is in the CONNECTED state). No other measurements should be running.

Example Program

```
10      ! INITIALIZE VARIABLES IN THE TEST
20      !
30      Sacch_time=TIMEDATE ! Set up a variable to capture test time
40      OPTION BASE 1 ! Set default lower bound of array subscripts to 1
50      REAL Integ1(100),Txp1(100),Valid_powers(301)
60      REAL Integ2(100),Txp2(100),Integ3(100),Txp3(100)
70      REAL Avg_results(20),Max_results(20),Min_results(20)
80      REAL Start_power,Stop_power
90      REAL Avg_pow_for_lvl,Sum_pow_for_lvl
100     REAL Max_power,Min_power
110     INTEGER Test_set,I,Start_level,Stop_level,Average_count,Power_count
120     Test_set=714
130     Start_level=5
140     Start_power=33.0
150     Stop_level=15
160     Average_count=0
170     Power_count=1
180     Max_power=0
190     Min_power=36
200     !
210     ! SETUP DPOW PARAMETERS
220     OUTPUT Test_set;"SET:DPOW:CONT OFF;EMD 2" ! Set the Expected Maximum Difference.
230     OUTPUT Test_set;"SET:DPOW:COUN:NUMBER 300" ! Set the number of bursts high enough
240                                     ! to capture all 10 power level changes
250     OUTPUT Test_set;"CALL:SIGN:MS:TXL:FACCH 1" ! Switch on FACCH mode
260     !
270     ! CHANGE POWER LEVEL FROM 5 TO 15
280     OUTPUT Test_set;"CALL:MS:TXL:SEQ ";Start_level
290     OUTPUT Test_set;"CALL:SIGN:MS:TXL:FACCH 0" ! Switch off FACCH signaling - only
300                                     ! SACCH header is used
310     OUTPUT Test_set;"CALL:MS:TXL:SEQ ";Stop_level;";:RFAN:EXP:POW ";Start_power
320     !
330     ! A SHORT WAIT MIGHT BE NEEDED HERE TO ENSURE 100 PERCENT RELIABILITY OF THE
340     ! MEASUREMENT. THIS MAY BE DEPENDENT ON THE PHONE AND IS TO ENSURE THAT THE VERY LAST
350     ! POWER STEP IS ALWAYS CAPTURED.
360     ! WAIT .1
370     !
380     ! INITIATE DPOW MEASUREMENT AND FETCH RESULTS
390     OUTPUT Test_set;"INIT:DPOW"
400     OUTPUT Test_set;"FETC:DPOW?" ! Query results for bursts 1 through 100
410     ENTER Test_set;Integ1(*),Txp1(*)
420     OUTPUT Test_set;"FETC:DPOW:RANG2?" ! Query results for bursts 101 through 200
430     ENTER Test_set;Integ2(*),Txp2(*)
440     OUTPUT Test_set;"FETC:DPOW:RANG3?" ! Query results for bursts 201 through 300
450     ENTER Test_set;Integ3(*),Txp3(*)
460     !
470     ! LOAD TXP RESULTS WHICH HAVE VALID INTEGRITY INTO ARRAY
480     !
490     ! PROCESS BURSTS 1 TO 100
500     K=1
510     FOR J=1 TO 100
520         IF Integ1(J)=0 THEN
530             Valid_powers(K)=Txp1(J)
540             K=K+1
550         END IF
560     NEXT J
570     ! PROCESS BURSTS 101 TO 200
580     FOR J=1 TO 100
590         IF Integ2(J)=0 THEN
600             Valid_powers(K)=Txp2(J)
610             K=K+1
620         END IF
630     NEXT J
640     ! PROCESS BURSTS 201 TO 300
650     FOR J=1 TO 100
660         IF Integ3(J)=0 THEN
670             Valid_powers(K)=Txp3(J)
680             K=K+1
690         END IF
```

Example Program (continued)

```
700 NEXT J
710 !
720 ! EXTRACT MIN, MAX AND AVERAGE POWER LEVELS FROM VALID RESULTS ARRAY
730 !
740 FOR I=1 TO 300
750   Average_count=Average_count+1
760   Sum_pow_for_lvl=Sum_pow_for_lvl+Valid_powers(I)
770   IF Valid_powers(I)>Max_power THEN
780     Max_power=Valid_powers(I)
790   END IF
800   IF Valid_powers(I)<Min_power THEN
810     Min_power=Valid_powers(I)
820   END IF
830   IF (Valid_powers(I)-Valid_powers(I+1))>=.8 THEN
840     Avg_pow_for_lvl=Sum_pow_for_lvl/Average_count
850     Avg_results(Power_count)=Avg_pow_for_lvl
860     Max_results(Power_count)=Max_power
870     Min_results(Power_count)=Min_power
880     Power_count=Power_count+1
890     Sum_pow_for_lvl=0
900     Average_count=0
910     Max_power=0
920     Min_power=36
930   END IF
940 NEXT I
950 I=1
960 ! PRINT MEASUREMENT RESULTS
970 CLEAR SCREEN
980 PRINT "Measured Power Control Levels from MS TX Level 5 to 15 using SACCH Method"
990 PRINT
1000 PRINT "MS TX Level   Average Power   Minimum Power   Maximum Power"
1010 PRINT "-----"
1020 FOR J=Start_level TO Stop_level
1030   PRINT J,"          ";PROUND(Avg_results(I),-3)," ";PROUND(Min_results(I),-3)," ";PROUND(Max_results(I),-3)
1040   I=I+1
1050 NEXT J
1060 PRINT
1070 ! PRINT TOTAL TIME FOR PROGRAM TO COMPLETE
1080 PRINT
1090 PRINT "Total Test Time: ";PROUND(TIMEDATE-Sacch_time,-2);"Seconds"
1100 END
```

Figure 2 shows a typical results screen that can be seen after running the example program.

Measured Power Control Levels from MS TX Level 5 to 15 using SACCH Method			
MS TX Level	Average Power	Minimum Power	Maximum Power
-----	-----	-----	-----
5	33.88	33.873	33.889
6	32.469	32.468	32.473
7	30.466	30.465	30.472
8	28.499	28.497	28.506
9	26.486	26.484	26.488
10	24.601	24.551	24.606
11	22.667	22.666	22.669
12	20.77	20.767	20.772
13	18.776	18.773	18.782
14	16.784	16.78	16.793
15	14.675	14.669	14.684

Total Test Time: 2.27 Seconds

Figure 2. Example results screen

Appendix A: Test set considerations when using SACCH MS TX Level Signaling

The following sections give you information concerning things you should know, unexpected behaviors, and potential problems that you may encounter when using SACCH MS TX Level signaling.

Things you should know

The front of the array holding the measured values will contain a high number of measurement results at the start_level setting

Given that performing an MS TX Level change on the SACCH will take 480 to 960 ms to receive the message, a high number of Dynamic Power measurements will be made before the power level starts to change in 2 dB steps. Consequently, there will be a lot of measurement results at the start_level setting (in this example 33 dBm) in the front portion of the storage array.

Note that the example program contains a WAIT statement at line 360 that is currently commented out. Depending on the particular mobile station you are testing, it may be necessary to remove the comment mark from this line and experiment with a short wait period to ensure that you can reliably measure all 2 dB power level steps.

Synchronizing the Dynamic Power measurement with each 2 dB power control level step

There is no synchronization between the mobile station changing power level and the triggering of the Dynamic Power measurement. The Dynamic Power measurement triggers on every burst whether or not the power level of that burst has changed from the previous burst measured.

Maintaining measurement accuracy while measuring power control level steps

The following are true for the mobile station and the test set:

- The power level changes in 2 dB steps
- AND
- If the input signal power is within ± 3 dB of the expected input power setting, the Dynamic Power measurement meets specified accuracy

The Expected Maximum Difference from the Previous Measurement parameter is used with the measured transmit power from the previous burst to set the maximum RF power that the base station emulator is expecting the mobile to transmit in the next burst. In the example program, the Expected Maximum Difference is set to 2 so that measurement accuracy is maintained when the mobile station executes a 2 dB step.

Potential problems

The first few measurement results in the result array are not at the initial MS TX Level setting

The following section is provided to enhance your understanding of a problem that could potentially occur when measuring power control levels. You should note that the example program does not suffer from this problem because it implements a solution presented below (where the expected power is set to the start power level). You only need to read this section if you want to further your understanding.

The following discussion refers to the program segment shown below:

```
90     REAL Integ1(100),Txp1(100)
100    REAL Integ2(100),Txp2(100),Integ3(100),Txp3(100)
110    OUTPUT Test_set;"SET:DPOW:EMD 2"  ! Set expected maximum
      difference
120    OUTPUT Test_set;"CALL:MS:TXL:SEQ 5"
130    OUTPUT Test_set;"CALL:SIGN:MS:TXL:FACCH 0"  ! Switch off
      FACCH signaling - only
140                                     ! SACCH header is used
150    OUTPUT Test_set;"CALL:MS:TXL:SEQ 15"
160    OUTPUT Test_set;"INIT:DPOW"
170    OUTPUT Test_set;"FETC:DPOW?"
180    ENTER Test_set;Integ1(*),Txp1(*)
190    OUTPUT Test_set;"FETC:DPOW:RANG2?"
200    ENTER Test_set;Integ2(*),Txp2(*)
210    OUTPUT Test_set;"FETC:DPOW:RANG3?"
220    ENTER Test_set;Integ3(*),Txp3(*)
```

Behavior of code

When this code is executed as shown, the first few elements of the Integ1 and Txp1 arrays (which hold the first 100 integrity and dynamic power results) may have values similar to the following:

```
5  18.10761  5  21.19696  5  24.25592  5  27.61498  5  30.73989 ...
```

The remainder of the Txp1 array will probably contain values corresponding to the starting level of 33 dB. The measured results of all the power steps will be contained in the Txp2 and Txp3 arrays, and 12 to 13 readings will be present in these arrays for each power step.

If line 150 in the code segment above is changed as follows:
(add;:RFAN:EXP:POW 33)

```
OUTPUT Test_set;"CALL:MS:TXL 15;:RFAN:EXP:POW 33"
```

The first few elements of the Txp1 and Integ1 arrays should always have values similar to the following:

```
0  33.08562  0  33.0779  0  33.07781  0  33.07737  0  33.07732 ...
```

Explanation of what is happening

- a) After execution of line 120 the expected input power = 33 dBm.
- b) After execution of line 150 the expected input power is in the process of being set to 13 dBm.
- c) Because of timing dependencies, the first power measurement is made AFTER the expected input power has been set to 13 dBm as a result of line 150. Since the mobile station's power is at 33 dBm and expected input power = 13 dBm, an overrange error (measurement integrity value = 5) is generated.
- d) The Expected Maximum Difference parameter (set on line 110) uses the value you set in conjunction with the measured transmit power from the previous burst to set the expected input power (that is, expected input power is set by the Expected Maximum Difference, input power = 33 dBm, this causes an overrange condition).
- e) The result is that you see the measured power APPEAR to ramp up at the beginning of the Txp1 array. Actually, the Expected Maximum Difference feature is ramping the expected input power setting to match the actual input power.

The test set leaves the MS TX Level to the expected input power coupling the same regardless of the state of the SACCH/FACCH signaling parameter. This is to ensure that the test set will behave in the same way, regardless of whether or not the test set used SACCH or SACCH and FACCH signaling for MS TX Level changes. If the MS TX Level to expected input power coupling had been disabled, it would always be necessary to set the expected input power appropriately. Forgetting to set the expected input power would result in measurement integrity errors and bad results as the signal's power went above or below the expected level setting.

When measuring mobile station power control levels, the control program can address this situation in one of two ways:

1. Accept the "overrange" integrity value that will sometimes occur for the first few measurements. This situation will exist until the expected maximum difference feature has set the expected input power to the actual transmit power. These can then be parsed out of the array based upon the measurement integrity value.
2. Immediately after setting MS TX Level to the stop level (which will automatically set expected input power to that level), set expected input power to the start power level (that is, the value that the mobile will be transmitting at, at that time) before continuing the program. The example program implements this approach.

Appendix B: A review of SACCH Messaging

Basic GSM frame structure

The GSM system utilizes time division multiplexing. A timeslot, the amount of time each mobile station has to transmit or receive information, has a time period of 576.92 uSec. There are 8 timeslots per frame and, for a traffic channel, 26 frames per multiframe. The GSM frame structure is shown in figure 3.

SACCH and FACCH signaling

GSM incorporates two bi-directional logical channels for the purpose of sending and receiving signaling information while the mobile station is on a traffic channel. A physical traffic channel is defined by an ARFCN (absolute radio frequency channel number) and a timeslot. Signaling information between the base station and the mobile station can be carried on either of the logical channels; the slow associated control channel (SACCH) or the fast associated control channel (FACCH).

SACCH signaling

Once per multiframe (see figure 3), one of the traffic channel's frames is used by the SACCH to carry signaling information rather than user data. On the downlink, the SACCH is used to send slowly but regularly changing control information to the mobile station, such as changes to MS TX Level or changes in burst timing advance. While on a traffic channel, the uplink SACCH carries measurement information such as received signal strength (RX Level) and received signal quality (RX Qual).




Figure 3. GSM frame structure

FACCH signaling

The FACCH is used in situations where large amounts of information must be communicated between the base station and the mobile station very fast, such as in a call handover. Communication is accomplished by replacing the data (in two chunks of 57 bits) with FACCH data for the time it takes to transfer the information. The FACCH uses the same ARFCN and timeslot assigned to the traffic channel. The FACCH effectively steals bursts from the traffic channel to communicate the information. When the FACCH steals bursts from the traffic channel, voice data is lost. However, the duration is very short and, in general, is not noticeable to the user.

SACCH message block

It takes four SACCH frames to carry one complete SACCH message block on the signaling channel. This requires the mobile station to store the contents of the SACCH frames, until four SACCH frames have been received. As shown in figure 4, it takes 480 ms to transmit a complete SACCH message block.




Figure 4. SACCH messaging timing

Appendix C: A review of mobile station power level control

Mobile station power level control refers to the mobile station's capability to change its transmit power level. Changing transmit power level is necessary to compensate for the attenuation of the transmitted signal over different distances within the cell and to conserve battery life. The base station commands the mobile station to change its transmit power level in such a way that the power arriving at the base station's receiver is approximately the same for each timeslot.

In the GSM system, the logical range over which a mobile station can change its transmit power is divided into 32 discrete steps, 0 to 31. Each step represents a relative 2 dB change. Maximum and minimum power levels vary based upon: a) mobile station power class, b) frequency band (PGSM, EGSM, DCS, PCS), and c) revision of GSM standards (Phase 1 or Phase 2). Commands to change transmit power level can be received on a SACCH message, or on a FACCH message (channel assignment or handover).

3GPP TS 05.08 section 4.7 version 8.9.0 Release 1999 states:

“Upon receipt of a command from [a] SACCH to change its power level on the corresponding uplink channel, the MS shall change to the new level at a rate of one nominal 2 dB power control step every 60 ms (13 TDMA frames), i.e. a range change of 15 steps should take about 900 ms. The change shall commence at the first TDMA frame belonging to the next reporting period (as specified in subclause 8.4).”

This behavior is illustrated in figure 5 for a transmit power level change from power control level (PCL) 5 to PCL 9 using SACCH signaling.




Figure 5. Mobile station power level control

Using SACCH messaging, the total time for the change from PCL5 to PCL9 will be 720 to 1200 ms. The variability is due to timing dependencies. If the command to change transmit power was issued just after the start of one SACCH message block, the command would have to wait until the remainder of the current SACCH block was transmitted before it could be included in the next transmitted SACCH block. The transmission of each SACCH message block is 480 ms. This messaging yields a possible maximum time of 960 ms (480 + 480 ms) to transmit the MS TX Level change.

Agilent Technologies' Test and Measurement Support, Services, and Assistance

Agilent Technologies aims to maximize the value you receive, while minimizing your risk and problems. We strive to ensure that you get the test and measurement capabilities you paid for and obtain the support you need. Our extensive support resources and services can help you choose the right Agilent products for your applications and apply them successfully. Every instrument and system we sell has a global warranty. Support is available for at least five years beyond the production life of the product. Two concepts underlie Agilent's overall support policy: "Our Promise" and "Your Advantage."

Our Promise

Our Promise means your Agilent test and measurement equipment will meet its advertised performance and functionality. When you are choosing new equipment, we will help you with product information, including realistic performance specifications and practical recommendations from experienced test engineers. When you use Agilent equipment, we can verify that it works properly, help with product operation, and provide basic measurement assistance for the use of specified capabilities, at no extra cost upon request. Many self-help tools are available.

Your Advantage

Your Advantage means that Agilent offers a wide range of additional expert test and measurement services, which you can purchase according to your unique technical and business needs. Solve problems efficiently and gain a competitive edge by contracting with us for calibration, extra-cost upgrades, out-of-warranty repairs, and on-site education and training, as well as design, system integration, project management, and other professional engineering services. Experienced Agilent engineers and technicians worldwide can help you maximize your productivity, optimize the return on investment of your Agilent instruments and systems, and obtain dependable measurement accuracy for the life of those products.

By internet, phone, or fax, get assistance with all your test & measurement needs

Online assistance:

www.agilent.com/find/assist

Phone or Fax

United States:

(tel) 1 800 452 4844

Canada:

(tel) 1 877 894 4414

(fax) (905) 282 6495

China:

(tel) 800 810 0189

(fax) 1 0800 650 0121

Europe:

(tel) (31 20) 547 2323

(fax) (31 20) 547 2390

Japan:

(tel) (81) 426 56 7832

(fax) (81) 426 56 7840

Korea:

(tel) (82 2) 2004 5004

(fax) (82 2) 2004 5115

Latin America:

(tel) (305) 269 7500

(fax) (305) 269 7599

Taiwan:

(tel) 080 004 7866

(fax) (886 2) 2545 6723

Other Asia Pacific Countries:

(tel) (65) 375 8100

(fax) (65) 836 0252

Email: tm_asia@agilent.com

Product specifications and descriptions in this document subject to change without notice.

© Agilent Technologies, Inc. 2001

Printed in USA September 18, 2001

5988-4002EN



Agilent Technologies